

RL-553 = RL-553

COMPUTER PROGRAM: RAYLEIGH SCATTERING

T. B. A. Senior and D. J. Ahlgren

The Radiation Laboratory
Department of Electrical and Computer Engineering
The University of Michigan
Ann Arbor, Michigan

1. INTRODUCTION

This program solves integral equations to compute the three independent polarizability tensor elements P_{11} , P_{33} and M_{11} , together with the capacitance C/ϵ and a quantity γ related to this for rotationally symmetric, perfectly conducting shapes composed of either one or two bodies. The tensor elements are as given in Keller et al (1972), but see also Senior and Ahlgren (1972) where the electromagnetic capacity C and the quantity γ are defined; ϵ is the permittivity of the surrounding medium. The numerical procedures are discussed and the various subroutines presented along with a flow chart of the program.

1.1 BODY DESCRIPTION AND ITS VOLUME

The body B is required to be rotationally-symmetric about the z -axis of the cylindrical coordinate system. The profile of the body in the ρ, z plane is given by the function $\rho = \rho(z)$, and is constructed of segments which may be linear arcs or circular arcs of either concavity. Distance along the profile is measured by the variable s while the tangent at any point forms the angle α with the z -axis; $-\pi/2 \leq \alpha \leq \pi/2$.

The volume of rotation of a linear arc (type 3) segment whose endpoints in the (z, ρ) plane are (z_1, ρ_1) , (z_2, ρ_2) is

$$\begin{aligned} \delta V_0 &= \int_{z_1}^{z_2} \pi \rho^2(z) dz \\ &= \frac{\pi}{3} (z_2 - z_1) (\rho_2^2 + \rho_1 \rho_2 + \rho_1^2) . \end{aligned} \quad (1)$$

For a circular arc which subtends an angle θ at its center (z_0, ρ_0) and is of radius a , define a quantity ξ where

$$\xi = \begin{cases} -1 & \text{segment concave down (type 1)} \\ 1 & \text{segment concave up (type 2).} \end{cases} \quad (2)$$

Then,

$$\delta V_0 = \pi \left| (z_2 - z_1) \left\{ \rho_0^2 + a^2 - \frac{1}{3} (u_1^2 + u_1 u_2 + u_2^2) \right\} - \xi \rho_0 \left\{ u_2 (\rho_2 - \rho_0) - u_1 (\rho_1 - \rho_0) + d a^2 \theta \right\} \right|, \quad (3)$$

where

$$d = \frac{z_2 - z_1}{|z_2 - z_1|} \quad (4)$$

and

$$u_2 = z_2 - z_0, \quad ,$$

$$u_1 = z_1 - z_0. \quad .$$

1.2 THE INTEGRAL EQUATIONS

Given the profile $\rho = \rho(z)$ and the volume V_0 , quantities m and m_1 are defined as

$$m = m(z, \rho, z', \rho') = \frac{4\rho\rho'}{(\rho + \rho')^2 + (z - z')^2} \quad (5)$$

and
$$m_1 = 1 - m \quad (6)$$

where (z, ρ) and (z', ρ') are distinct points on the profile $\rho(z)$, the unprimed variables denoting an observation point and the primed variables a remote point. Note that $0 \leq m, m_1 \leq 1$.

Denote by $K(m)$, $E(m)$ respectively, the complete elliptic integrals of the

first and second kinds, and let $K'(m) = \frac{d}{dm} K(m)$. Then follow the steps listed below:

- i. To compute P_{11}/V_0 , solve

$$\int_0^s \rho' K_1 T_1(s') ds' = 2\pi\rho \quad (7)$$

where $K_1 = K_1(z, \rho, z', \rho')$

$$= \frac{2}{(m\rho\rho')^{1/2}} \left\{ \left(1 - \frac{m}{2}\right) K(m) - E(m) \right\} \quad (8)$$

and compute

$$\frac{P_{11}}{V_0} = \frac{\pi}{V_0} \int_0^s \rho^2 T_1(s) ds \quad . \quad (9)$$

- ii. To compute P_{33}/V_0 , C/ϵ , and γ , solve

$$\int_0^s \rho' K_0 T_2(s') ds' = 2\pi z \quad , \quad (10)$$

and
$$\int_0^s \rho' K_0 T_3(s') ds' = 2\pi \quad (11)$$

where
$$K_0 = \left(\frac{m}{\rho\rho'} \right)^{1/2} K(m) \quad , \quad (12)$$

$$C/\epsilon = 2\pi \int_0^s \rho T_3(s) ds \quad , \quad (13)$$

$$\gamma = -\frac{\epsilon}{C} 2\pi \int_0^s z \rho T_3(s) ds \quad (14)$$

and

$$\frac{P_{33}}{V_0} = \frac{2\pi}{V_0} \int_0^s z \rho T_2(s) ds - \frac{C}{\epsilon} \frac{\gamma^2}{V_0} \quad (15)$$

iii. If and only if B consists of two separate (electrically unconnected) closed parts B_1 and B_2 , solve

$$\int_0^s \rho' K_0 T_3^{(1)}(s') ds' = \begin{cases} 2\pi & \text{for points on } B_1 \\ 0 & \text{for points on } B_2 \end{cases} ; \quad (16)$$

compute

$$\frac{\delta P_{33}}{V_0} = \frac{-\frac{2\pi}{V_0} \left\{ \int_0^s (z + \gamma) \rho T_3^{(1)}(s) ds \right\}^2}{\int_{(1)} \rho T_3^{(1)}(s) ds - \frac{\epsilon}{C} 2\pi \left\{ \int_{(1)} \rho T_3(s) ds \right\}^2}, \quad (17)$$

where the symbol (1) below the integral sign means that the integration is carried out over the profile of B_1 alone.

Then the tensor element P_{33} for the disjoint shape is

$$\frac{\tilde{P}_{33}}{V_0} = \frac{P_{33}}{V_0} + \frac{\delta P_{33}}{V_0} \quad (18)$$

iv. Solve

$$\begin{aligned} \bar{\int}_0^s V_4(s') \left\{ \rho \cos \alpha' \Omega_2 + \left[(z' - z) \sin \alpha' - \rho' \cos \alpha' \right] \Omega_1 \right\} \rho' ds' \\ = \pi \left\{ V_4(s) - 2\rho \right\} \end{aligned} \quad (19)$$

where

$$\Omega_1 = \left(\frac{m}{\rho \rho'} \right)^{3/2} \left\{ \left(1 - \frac{m}{2} \right) K'(m) - \frac{1}{4} K(m) \right\} , \quad (20)$$

$$\Omega_2 = \frac{1}{m^2} \left(\frac{m}{\rho \rho'} \right)^{3/2} \left\{ 2m \left(1 - \frac{m}{2} \right)^2 K'(m) - \left(1 - \frac{m^2}{4} \right) K(m) + E(m) \right\} \blacksquare \quad (21)$$

and the bar across the integral sign denotes the Cauchy principal value.

Compute

$$\frac{M_{11}}{V_0} = \frac{\pi}{V_0} \int_0^s \rho V_4(s) \cos \alpha ds . \quad (22)$$

2. NUMERICAL SOLUTION OF THE INTEGRAL EQUATIONS

The numerical procedures involved in finding P_{33}/V_0 , M_{11}/V_0 and where appropriate, \tilde{P}_{33}/V_0 are quite similar to those required for P_{11}/V_0 , and it is therefore sufficient to give full details only for P_{11}/V_0 .

2.1 P_{11}/V_0 COMPUTATION

The primary task is the solution of the integral equation (7) for the function $T_1(s)$ and this entails the determination of a sequence of values $T_1^{(i)}$, $i = 1, 2, \dots, N$, approximating $T_1(s)$ at the sampling points $s = s_i$ on the profile $\rho = \rho(z)$. For this purpose the profile is divided into N cells C_i of arc length Δs_i and midpoints s_i corresponding to the coordinates (ρ_i, z_i) . Within each cell we also define the points s_{i-} and s_{i+} where

$$s_{i-} = s_i - \alpha_0 \Delta s_i \tag{23}$$

$$s_{i+} = s_i + \alpha_0 \Delta s_i$$

with the restriction

$$0 \leq \alpha_0 \leq \frac{1}{2} .$$

By assuming that $T_1(s)$ has the constant value $T_1^{(i)}$ over the i th cell,

the integral on the left hand side of (7) can be evaluated as a linear combination of the $T_1^{(i)}$ whose coefficients depend on the position (ρ, z) of the field point, leading to a linear system of N equations in N unknowns, viz.

$$T_1^{(1)} \int_{C_1} \rho' K_1 ds' + T_1^{(2)} \int_{C_2} \rho' K_1 ds' + \dots + T_1^{(N)} \int_{C_N} \rho' K_1 ds' = 2\pi\rho_i$$

$$i = 1, 2, \dots, N. \quad (24)$$

Hence, the system to be solved is

$$A t_1 = b \quad (25)$$

where t_1 is a column vector with elements

$$t_{1i} = T_1^{(i)}, \quad i = 1, 2, \dots, N, \quad (26)$$

A is a square matrix with elements

$$a_{ij} = \int_{C_j} \rho' K_1 ds', \quad i, j = 1, 2, \dots, N, \quad (27)$$

and b is a row vector with elements

$$b_j = 2\pi\rho_j, \quad j = 1, 2, \dots, N. \quad (28)$$

Increasing the complexity of the quadrature technique used to evaluate the integrals \int_{C_j} will generally improve the accuracy but will almost

certainly increase the computational cost. What is therefore desired is the least expensive procedure capable of giving the required accuracy. The two simplest approaches are to integrate first and second order approximations to give ($i \neq j$):

$$a_{ij} = \rho_j K_1(i, j) \Delta s_j \tag{29}$$

$$a_{ij} = \left[w_1(\alpha_0) \left\{ \rho_{j-} K_1(i, j-) + \rho_{j+} K_1(i, j+) \right\} + w_0(\alpha_0) \rho_j K_1(i, j) \right] \Delta s_j \tag{30}$$

respectively, where the subscripts $j-$ and $j+$ correspond to the points s_{j-} and s_{j+} of eqs. (23), and $K_1(i, j)$ is the kernel defined in eq. (8) and evaluated at the points $(\rho_i, z_i), (\rho_j, z_j)$. By requiring $\alpha_0 < \frac{1}{2}$, we ensure that the sampling points s_{j-} and s_{j+} do not coincide with the end points of the cell C_j , and thereby avoid any difficulty in the computations of Ω_1 and Ω_2 (see eqs. 20 and 21). When

$$\alpha_0 = \frac{1}{2} \sqrt{\frac{3}{5}} \tag{31}$$

eq. (30) reduces to the three-point Gaussian formula for which

$$w_0 = \frac{4}{9} \quad , \quad w_1 = \frac{5}{18} \tag{32}$$

With this choice of w_0 and w_1 , the advantages of eq. (30) vis-a-vis eq. (29) were now determined by computing P_{11}/V_0 for a sphere using various values of N . Fig. 1 shows percent accuracy and C. P. U. time versus N for each integration scheme. It is apparent that for a given expenditure of C. P. U. time the Gaussian three-point technique is much more accurate than




Fig. 1: Percent error and C. P. U. time of P_{11}/V_0 calculation for a sphere: T denotes trapezoidal rule computation and G denotes three-point Gaussian.

the trapezoidal method, though the accuracies of both are severely degraded if N is too small ($N \lesssim 5$). Since the Gaussian scheme with $N = 10$ produces an accuracy of better than 99.8 percent for a sphere, there is no point in going to a more complicated procedure, and the computer program was therefore written using three-point Gaussian quadrature to determine the matrix elements a_{ij} .

In summary, the integral equation (7) is solved by conversion to the matrix system (25) in which

$$a_{ij} = \left[\frac{5}{18} \left\{ \rho_{j-} K_1(i, j-) + \rho_{j+} K_1(i, j+) \right\} + \frac{4}{9} \rho_j K_1(i, j) \right] \Delta s_j$$

$$i, j = 1, 2, \dots, N; i \neq j \quad (33)$$

$$a_{ii} = \left[\ln \left(\frac{16\rho_i}{\Delta s_i} \right) - 1 \right] \Delta s_i, \quad i = 1, 2, \dots, N.$$

Having determined the sampled values $T_1^{(i)} = T_1(s_i)$, P_{11}/V_0 is computed from eq. (9) by integration over each segment of the profile using a second order integration procedure (subroutine INTEG, described in Section 3.4).

2.2 P_{33}/V_0 COMPUTATION

The point sampling method of solution of the integral equations (10) and (11) requires us to find the sequences $T_2^{(i)} = T_2(s_i)$ and $T_3^{(i)} = T_3(s_i)$, $i = 1, 2, \dots, N$, from these equations. To determine the $T_2^{(i)}$, choose α_0 , w_0 and w_1 in accordance with (31) and (32) and thence solve the matrix

system $At_2 = b$ where

$$t_{2i} = T_2^{(i)} \quad i = 1, 2, \dots, N \quad (34)$$

$$b_i = 2\pi z_i$$

and

$$a_{ij} = \left[\frac{5}{18} \left\{ \rho_{j-} K_0(i, j-) + \rho_{j+} K_0(i, j+) \right\} + \frac{4}{9} \rho_j K_0(i, j) \right] \Delta s_j$$

$$i, j = 1, 2, \dots, N; i \neq j, \quad (35)$$

$$a_{ii} = \left[\ln \left(\frac{16\rho_i}{\Delta s_i} \right) + 1 \right] \Delta s_i, \quad i = 1, 2, \dots, N.$$

The $T_3^{(i)}$ are similarly determined by solving the matrix system $At_3 = b$ where the elements a_{ij} are again given by (35), but

$$t_{3i} = T_3^{(i)} \quad i = 1, 2, \dots, N \quad (36)$$

$$b_i = 2\pi.$$

The quantities C/ϵ , γ and P_{33}/V_0 defined in eqs. (13), (14) and (15)

respectively are computed using the same second order integration procedure employed in calculating P_{11}/V_0 .

If the body profile consists of two discrete parts, it is also necessary to solve the integral equation (16). The corresponding matrix system is

almost identical to that in (36), and from the sampled values $T_3^{(1)}(s_i)$ and $T_3(s_i)$, $\delta P_{33}/V_0$ (see eq. 17) is computed and, hence, \tilde{P}_{33}/V_0 .

2.3 M_{11}/V_0 COMPUTATION

The basic approach is similar to the above in spite of the more complicated integral equation (19) that must now be solved. The matrix equation for the sampled values $V_4(s_i) = V_4^{(i)}$ is $A v_4 = b$ where

$$v_{4i} = V_4^{(i)} \quad i = 1, 2, \dots, N \quad (37)$$

$$b_i = 2\pi\rho_i$$

and

$$a_{ij} = - \left[\frac{5}{18} \left\{ \rho_{j-} f(i, j-) + \rho_{j+} f(i, j+) \right\} + \frac{4}{9} f(i, j) \right] \Delta s_j \quad i, j = 1, 2, \dots, N; \quad i \neq j, \quad (38)$$

$$a_{ii} = \pi - \int_{s_i - \frac{1}{2} \Delta s_i}^{s_i + \frac{1}{2} \Delta s_i} \rho' f(s') ds' \quad i = 1, 2, \dots, N$$

in which

$$f(i, j) = \left[\rho_i \cos \alpha_j \Omega_2(i, j) + \left\{ (z_j - z_i) \sin \alpha_j - \rho_j \cos \alpha_j \right\} \Omega_1(i, j) \right] \quad i, j = 1, 2, \dots, N; \quad i \neq j. \quad (39)$$

We observe that the computation of each diagonal element of A requires the numerical evaluation of a Cauchy principal value (denoted by the bar across the integral sign in the above expression for a_{ii}). As an approximation to this principal value, we remove from the cell C_i a slice defined by the interval $(s_i - \frac{1}{2} \beta \Delta s_i, s_i + \frac{1}{2} \beta \Delta s_i)$ where $\beta, 0 < \beta \leq 1$, is the fractional exclusion; $\beta = 1$ implies no exclusions, i. e. that the principal value is not taken.

We now have

$$a_{ii} \approx \pi - \int_{s_i - \frac{1}{2} \Delta s_i}^{s_i - \frac{\beta}{2} \Delta s_i} \rho' f(s') ds' - \int_{s_i + \frac{\beta}{2} \Delta s_i}^{s_i + \frac{1}{2} \Delta s_i} \rho' f(s') ds' \quad (40)$$

and these integrals are also computed using three-point Gaussian quadrature.

Defining

$$\begin{aligned} s_{i2} &= s_i - \frac{1}{4} (1+\beta) \Delta s_i \\ s_{i1} &= s_{i2} - \frac{1}{2} \alpha_0 (1-\beta) \Delta s_i \\ s_{i3} &= s_{i2} + \frac{1}{2} \alpha_0 (1-\beta) \Delta s_i \\ s_{i5} &= s_i + \frac{1}{4} (1+\beta) \Delta s_i \\ s_{i4} &= s_{i5} - \frac{1}{2} \alpha_0 (1-\beta) \Delta s_i \\ s_{i6} &= s_{i5} + \frac{1}{2} \alpha_0 (1-\beta) \Delta s_i \end{aligned} \quad (41)$$

we obtain

$$a_{ii} \approx \pi - \frac{1}{2}(1 - \beta) \left[\frac{5}{18} \left\{ \rho_{i1} f(i, i_1) + \rho_{i3} f(i, i_3) + \rho_{i4} f(i, i_4) + \rho_{i6} f(i, i_6) \right\} + \frac{4}{9} \left\{ \rho_{i2} f(i, i_2) + \rho_{i5} f(i, i_5) \right\} \right] \Delta s_i . \quad (42)$$

Equations (37) through (42) completely describe a system of N linear equations in N unknowns V_{4i} , $i = 1, 2, \dots, N$. Their solution and subsequent integration of the V_{4i} according to eq. (22) yield M_{11}/V_0 .

3. THE COMPUTER PROGRAM

The program computes P_{11}/V_0 , C/ϵ , γ , P_{33}/V_0 , M_{11}/V_0 and, where appropriate, \tilde{P}_{33}/V_0 , and consists of a main program and six subroutines.

3.1 DATA SET

A data set is made up of one control card and a number of segment specification cards, one for each segment (or sub-segment) of the profile.

Control Card

Columns	Description
1	The number (1 or 2) of bodies.
3-4	Two digit integer (right justified): the number of segments on the first body (the body to the left). When there is only one body, use these columns.
6-7	Same as columns 3-4, but for body to the right.
9	A printing key: 1: print T_3 from P_{33}/V_0 computation. 0 or blank: do not print T_3 .

11	A computation key (0, blank or 1) 1: suppresses computation of P_{11}/V_0 , C/ϵ , γ , P_{33}/V_0 .
13	A computation key (0, blank or 1) 1: suppresses computation of M_{11}/V_0 .
21-30	A real number: the fractional exclusion β . If these columns are blank, β defaults to 0.001.

Segment Specification Card

Columns	Description
1-2	Two digit integer (right justified): the number of sampling points or cells on the segment.
4	Segment type key: 1: circular arc, concave down 2: circular arc, concave up 3: linear.
6	Volume sense: + or blank: additive volume - : subtractive volume.
11-20, 21-30	Two real numbers: respectively, the end coordinates z_1 and z_2 of the segment.

31-40, 41-50

Two real numbers: respectively, the end coordinates ρ_1 and ρ_2 of the segment.

51-60

A real number: for circular arcs, the included angle in degrees.


There are the following restrictions:

- (i) the total number of segments must not exceed 15,
- and (ii) the total number of cells over all segments must not exceed 80.

The profile is specified in the direction of increasing profile-length, beginning at its left-hand intersection with the z-axis and ending at its right-hand intersection with the z-axis. Re-entrant segments are permitted, allowing $z_1 > z_2$.

3.2 MAIN PROGRAM

The main program reads and prints data and supervises all computations. A rough flow chart showing the interaction of the subroutines is given below



```

REAL AP11(80,80),AP33(80,80),AM11(80,80),X(80),B(80),
6 ZEP(2),RHOEP(2),ST3(6),T3(80),M11,T1(80)
INTEGER NUMPTS(15),PLUS/'+'/,RL/' '/,INDX(2)
COMMON RHO(80,9),Z(80,9),ARC(80),C(80,9),S(80,9)/SUL/IPS(80)
DATA MIN,TWOPI,PI,W0,W1/'- ',6.283185,3.141593,.4444444,.2777777/
37 READ(5,34,END=999)NBOD,NS1,NS2,IPRINT,KEYP11,KEYM11,FR
34 FORMAT(I1,1X,2(I2,1X),3(I1,1X),6X,F12.7)
WRITE(6,4)NBOD,NS1
4 FORMAT('1*** BEGINNING OF DATA SET:/'/'0',5X,'BODIES',6X,'='',I2/
5 ' ',5X,'SEGMENTS:/'/' ',5X,'BODY #1',5X,'='',I2)
IF(NBOD .LE. 0 .OR. NBOD .GT. 2) GO TO 990
NSEGS=NS1+NS2
IF(NBOD .EQ. 2) WRITE(6,1)NS2
1 FORMAT(' ',5X,'BODY #2',5X,'='',I2)
IF(NBOD .EQ. 1) GO TO 10
IF(NS1 .LE. 0 .OR. NS2 .LE. 0) GO TO 990
10 IF(NSEGS .LE. 0 .OR. NSEGS .GT. 15) GO TO 990
WRITE(6,352)IPRINT,KEYP11,KEYM11
352 FORMAT(' ',5X,'PRINT KEY',3X,'='',I2/' ',5X,'COMP KEY P''S=',I2/
6 ' ',5X,'COMP KEY M''S=',I2)
IF(FR .LE. 0 .OR. FR .GT. 1.) FR=.001
IF(KEYM11 .EQ. 0) WRITE(6,5)FR
5 FORMAT(' ',5X,'EXCLUSION =',F7.4)
IF(KEYP11 .NE. 0 .AND. KEYM11 .NE. 0) GO TO 990
M=0
NC1=0
V0=0.0
DO 11 I=1,NSEGS
READ(5,12) NUMPTS(I),ITYP,ISIGN,ZEP,RHOEP,THETA
12 FORMAT(I2,1X,I1,1X,A1,4X,5F10.7)
IF(NUMPTS(I) .LE. 0 .OR. ITYP .LE. 0 .OR. ITYP .GT. 3) GO TO 990
IF(ISIGN .EQ. RL) ISIGN=PLUS
WRITE(6,13)I,NUMPTS(I),ITYP,ISIGN,ZEP,RHOEP
13 FORMAT('0SEGMENT #',I2,':/'/' ',5X,'CELLS',7X,'='',I2/' ',5X,
6 'TYPE KEY',4X,'='',I2/' ',5X,'VOLUME SENSE=',A4/' ',5X,
8 'Z-COORDINATE END POINTS =('',F12.7,',',',',F12.7,',')/'/' ',5X,
5 'RHO-COORDINATE END POINTS=('',F12.7,',',',',F12.7,',')')
IF(ITYP .NE. 3) WRITE(6,14) THETA
14 FORMAT(' ',5X,'THETA (DEG) =',F10.5)
IF(KEYM11 .EQ. 1) FR=1.0
N=M
THETA=PI*THETA/180.
M=M+NUMPTS(I)
IF(I .LE. NS1) NC1=NC1+NUMPTS(I)
IF(M .GT. 80) GO TO 990
CALL DATA(ITYP,N,M,ZEP,RHOEP,THETA,FR,VOLINC)
IF(ITYP .NE. 3 .AND. ISIGN .EQ. MIN) VOLINC=-VOLINC
11 V0=V0+VOLINC
WRITE(6,52)V0
52_ FORMAT('0COMPUTED RESULTS:/'/' ',5X,'VOLUME',6X,'='',F10.5)

```

```

      DO 2 N=1,M
      INDX(1)=N
      AN=ARC(N)
      TN=RHO(N,8)
      DO 3 L=N,M
      IF(L .EQ. N) GO TO 82
      AL=ARC(L)
      TL=RHO(L,8)
      IF(KEYM11-1)110,109,109
110  AM11(N,L)=0.0
      AM11(L,N)=0.0
109  IF(KEYP11-1)111,112,112
111  AP11(N,L)=0.0
      AP11(L,N)=0.0
      AP33(N,L)=0.0
      AP33(L,N)=0.0
112  INDX(2)=L

```

```

DO 103 J=1,3,2
JP6=J+6
DO 104 LL=1,2
I=3-LL
I1=INDX(LL)
I2=INDX(I)
TI2=RHO(I2,JP6)
CALL SETUP(KEYP11,KEYM11,I1,I2,JP6,API11,API33,AMI11,1)
IF(KEYP11-1)105,106,106
105 AP11(I1,I2)=AP11(I1,I2)+API11*TI2
AP33(I1,I2)=AP33(I1,I2)+API33*TI2
106 IF(KEYM11-1)107,104,104
107 AM11(I1,I2)=AM11(I1,I2)-AMI11*TI2
104 CONTINUE
103 CONTINUE
CALL SETUP(KEYP11,KEYM11,N,L,8,API11,API33,AMI11,1)
IF(KEYP11-1)108,209,209
108 U=W0*API11
AP11(N,L)=AL*(W1*AP11(N,L)+U*TL)
AP11(L,N)=AN*(W1*AP11(L,N)+U*TN)
U=W0*API33
AP33(N,L)=AL*(W1*AP33(N,L)+U*TL)
AP33(L,N)=AN*(W1*AP33(L,N)+U*TN)
209 IF(KEYM11-1)210,3,3
210 AM11(N,L)=AL*(W1*AM11(N,L)-AMI11*W0*TL)
CALL SETUP(1,0,L,N,8,API11,API33,AMI11,0)
AM11(L,N)=AN*(W1*AM11(L,N)-AMI11*W0*TN)
GO TO 3
82 IF(KEYP11-1)83,84,84
83 U=ALOG(16.*TN/AN)
AP11(N,N)=(U-1.0)*AN
AP33(N,N)=(U+1.0)*AN
84 IF(KEYM11-1)85,3,3
85 IF(FR .EQ. 1.0) GO TO 3
DO 86 I=1,6
86 CALL SETUP(1,0,N,N,I,U,U,ST3(I),1)
U=.5*(1.0-FR)*AN
AM11(N,N)=PI-U*(W0*(RHO(N,2)*ST3(2)+RHO(N,5)*ST3(5))
3 +W1*(RHO(N,1)*ST3(1)+RHO(N,3)*ST3(3)+RHO(N,4)*ST3(4)+
4 RHO(N,6)*ST3(6)))
3 CONTINUE
2 CONTINUE

```

```

      DO 20 I=1,M
20    R(I)=TWOPI*RHO(I,8)
      IF(KEYP11-1)21,24,24
21    CALL DECOMP(AP11,M)
      CALL SOLVE(AP11,X,R,M)
      DO 22 I=1,M
22    X(I)=RHO(I,8)**2*X(I)
      CALL INTEG(X,NSEGS,NUMPTS,P11)
      P11=P11*PI/VO
24    IF(KEYM11-1)25,28,28
25    CALL DECOMP(AM11,M)
      CALL SOLVE(AM11,X,R,M)
      DO 26 I=1,M
26    X(I)=RHO(I,8)*C(I,8)*X(I)
      CALL INTEG(X,NSEGS,NUMPTS,M11)
      M11=PI*M11/VO
28    IF(KEYP11-1)32,45,45
32    DO 29 I=1,M
29    R(I)=TWOPI*Z(I,8)
      CALL DECOMP(AP33,M)
      CALL SOLVE(AP33,X,R,M)
      DO 134 I=1,M
      X(I)=Z(I,8)*RHO(I,8)*X(I)
134   R(I)=TWOPI
      CALL INTEG(X,NSEGS,NUMPTS,P33)
      CALL SOLVE(AP33,T3,R,M)
      DO 35 I=1,M
      X(I)=RHO(I,8)*T3(I)
35    R(I)=Z(I,8)*X(I)
      CALL INTEG(X,NSEGS,NUMPTS,CAP)
      CAP=TWOPI*CAP
      CALL INTEG(R,NSEGS,NUMPTS,GAM)
      GAM=-TWOPI*GAM/CAP
      P33=(TWOPI*P33-CAP*GAM*GAM)/VO
      IF(NBOD - 1) 39,39,54
54    DO 36 I=1,M
      IF(I-NC1)305,305,306
305   R(I)=TWOPI
      GO TO 36
306   R(I)=0.0
36    CONTINUE

```

```

      CALL SOLVE(AP33,T1,R,M)
      DO 307 I=1,M
307   X(I)=RHO(I,8)*T1(I)
      CALL INTEG(X,NSEGS,NUMPTS,TL)
      CALL INTEG(X,NS1,NUMPTS,TN)
      DO 308 I=1,M
308   X(I)=Z(I,8)*X(I)
      CALL INTEG(X,NSEGS,NUMPTS,U)
      DELTAP=- (TWOPI/VO)*(U+GAM*TL)**2/(TN-TWOPI*TL*TL/CAP)
      U=P33+DELTAP
39   WRITE(6,40)CAP,GAM,P11,P33
40   FORMAT(' ',5X,'CAPACITANCE =',F10.5/' ',5X,'GAMMA',7X,'=',F10.5/
8   ' ',5X,'P11/V',7X,'=',F10.5/' ',5X,'P33/V',7X,'=',F10.5)
      IF(NBDD .EQ. 2)WRITE(6,309)DELTAP,U
309   FORMAT(' ',5X,'DELT P33/V',2X,'=',F10.5/' ',5X,'DISJNT P33/V=',
6   F10.5)
45   IF(KEYM11-1)42,337,337
42   WRITE(6,43)M11
43   FORMAT(' ',5X,'M11/V',7X,'=',F10.5)
337   IF(IPRINT .EQ. 1 .AND. KEYP11 .EQ. 0) WRITE(6,44)
44   6 (Z(I,8),RHO(I,8),T3(I),I=1,M)
      FORMAT('0',5X,'Z',10X,'RHO',12X,'T3'/' ',3(F12.6,2X))
      GO TO 37
990   WRITE(6,991)
991   FORMAT('0*** ERROR IN DATA')
999   CALL SYSTEM
      END

```

3.3 SUBROUTINE DATA (IN, MX, MY, ZEP, RHOEP, THETA, B, VOL)

This subroutine is called once for each segment of the profile. From the input specification for the segment, DATA computes the (z, ρ) coordinates of the necessary sampling points on the profile, the quantities $\cos \alpha$ and $\sin \alpha$ at these points and the incremental volume of the segment.

Arguments :

IN	Type key for segment.
MX	Total number of cells in segments to the left.
MY	MX + (number of cells in this segment).
ZEP	z - coordinate end points of segment : ZEP (1) = z_1 , ZEP (2) = z_2 .
RHOEP	ρ - coordinate end points of segment : RHOEP (1) = ρ_1 , RHOEP (2) = ρ_2 .
THETA	Angle (in radians) subtended by a circular arc at its center.
B	Fractional exclusion, β .
VOL	Incremental volume of segment.

Comments :

Stored in COMMON are the arrays RHO (80, 9), Z(80, 9), ARC (80), C(80, 9) and S (80, 9) which contain the numbers computed by DATA .

For the Ith cell, the subscripts (I, J) correspond to the points s_{ij} of (41) when $1 \leq J \leq 6$. For $J = 7, 8, 9$, the subscripts (I, J) refer to the points s_{i-} , s_i and s_{i+} respectively of (23).


```

SUBROUTINE DATA(IN,MX,MY,ZEP,RHOEP,THETA,B,VOL)
DIMENSION ZEP(2),RHOEP(2)
COMMON RHO(80,9),Z(80,9),ARC(80),C(80,9),S(80,9)
DATA STEP/.3872988/
MXP1=MX+1
EN=FLOAT(MY-MX)
IF(B .NE. 1) SUBSTP=.5*(1.0-B)*STEP
IF(IN-2)1,2,3
1   CC=-1.0
   GO TO 10
2   CC=1.0
10  ST2=SIN(THETA/2.0)
   A=ZEP(2)-ZEP(1)
   RAD=0.5*SQRT((RHOEP(1)-RHOEP(2))**2+A*A)/ST2
   DD=A/ABS(A)
   T=CC*DD*COS(THETA/2.)/ST2
   ZCNT=0.5*(ZEP(1)+ZEP(2)+T*(RHOEP(1)-RHOEP(2)))
   RHOCNT=0.5*(RHOEP(1)+RHOEP(2)+T*A)
   U2=ZEP(2)-ZCNT
   U1=ZEP(1)-ZCNT
   VOL=3.141593*ABS(A*(RHOCNT**2+RAD*RAD-(U2**2+U1*U2+U1**2)/3.0)
3  -CC*RHOCNT*(U2*(RHOEP(2)-RHOCNT)-U1*(RHOEP(1)-RHOCNT) +RAD*RAD
3  *DD*THETA))
   BETA=CC*DD*THETA/EN
   THET1=ATAN2(RHOEP(1)-RHOCNT,ZEP(1)-ZCNT)
   U=ABS(BETA*RAD)
   B3=STEP*BETA
   DO 902 I=MXP1,MY
   PHI=THET1+(I-MX-.5)*BETA
   IF(B .EQ. 1.0) GO TO 1905
   DO 1902 J=1,2
   ANG=PHI+.5*(J-1.5)*BETA*(1.0+B)
   DO 1903 L=1,3
   PSI=ANG+(L-2)*SUBSTP*BETA
   M=L+3*(J-1)
   C(I,M)=-CC*SIN(PSI)
   S(I,M)=CC*COS(PSI)
   Z(I,M)=ZCNT+RAD*CC*S(I,M)
1903 RHO(I,M)=RHOCNT-CC*RAD*C(I,M)
1902 CONTINUE
1905 DO 903 J=7,9
   ANG=PHI+(J-8)*B3
   C(I,J)=-CC*SIN(ANG)
   S(I,J)=CC*COS(ANG)
   Z(I,J)=ZCNT+RAD*CC*S(I,J)
903  RHO(I,J)=RHOCNT-CC*RAD*C(I,J)
902  ARC(I)=U
   RETURN

```

```

3      DX=(ZEP(2)-ZEP(1))/EN
      DY=(RHOEP(2)-RHOEP(1))/EN
      U  =SQRT(DX*DX+DY*DY)
      SI=DY/U
      CI=DX/U
      DO 917 I=MXP1,MY
      PHI=FLOAT(I-MX)-.5
      IF( B .EQ. 1.0) GO TO 1800
      DO 1802 J=1,2
      ANG=PHI+.5*(J-1.5)*(1.0+B)
      DO 1803 L=1,3
      M=L+3*(J-1)
      PSI=ANG+(L-2)*SUBSTP
      Z(I,M)=ZEP(1)+PSI*DX
      RHO(I,M)=RHOEP(1)+PSI*DY
      S(I,M)=SI
1803   C(I,M)=CI
1802   CONTINUE
1800   DO 913 J=7,9
      ANG=PHI+(J-8)*STEP
      Z(I,J)=ZEP(1)+ANG*DX
      RHO(I,J)=RHOEP(1)+ANG*DY
      C(I,J)=CI
913    S(I,J)=SI
917    ARC(I)=U
      VOL=1.047198*(ZEP(2)-ZEP(1))*(RHOEP(1)**2+RHOEP(1)*RHOEP(2)+
8 RHOEP(2)**2)
      RETURN
      END

```

3.4 SUBROUTINE INTEG (V, NSEG, NUMPTS, SUM)

INTEG numerically integrates quadratic interpolating polynomials approximating the data on each segment of the profile. When the profile is composed of several segments, no interpolation is performed across segment boundaries. Hence, the integration is accurate even for disconnected segments, e.g. the circular arcs of two spheres.

Arguments :

V	Real vector of function values, ordered as the cells.
NSEG	Total number of segments in the profile.
NUMPTS	Integer array containing in NUMPTS (I) the number of cells on the Ith segment: $I = 1, \dots, \text{NSEG}$.
SUM	Integral of V across the profile.

Comments :

Stored in COMMON are the arc lengths $\text{ARC}(I)$, $I = 1, \dots, N$ required to compute the integral.

```
SUBROUTINE INTEG(V,NSEG,NUMPTS,SUM)
COMMON RHO(80,9),Z(80,9),ARC(80),C(80,9),S(80,9)
DIMENSION V(80),NUMPTS(15)
SUM=0.0
JACC=1
DO 3000 I=1,NSEG
  T=ARC(JACC)
  L=NUMPTS(I)
  N=I+JACC-1
  SUM=SUM+T*(0.625*(V(JACC)+V(N))-0.125*(V(JACC+1)+V(N-1)))
  IF(L/2 .NE. (L+1)/2) GO TO 3001
  SUM=SUM+T*(0.6666667*V(N-1)-0.08333333*V(N-2)+0.4166667*V(N))
3001  LM1=N-1
      JLD=JACC+1
      DO 3002 J=JLD,LM1,2
3002  SUM=SUM+0.3333333*T*(V(J-1)+4.0*V(J)+V(J+1))
3000  JACC=JACC+L
      RETURN
      END
```

3.5 SUBROUTINES DECOMP (A, N) AND SOLVE (A, X, B, N)

Used together, DECOMP and SOLVE solve the linear system $AX = B$. DECOMP performs a L-U decomposition of the $N \times N$ matrix A and SOLVE performs back-substitution. These routines are adapted from Forsythe and Moler (1967, pp. 68-69).

```

SUBROUTINE DECOMP(UL,N)
DIMENSION UL(80,80)
COMMON /SOL/IPS(80)
DO 5 I=1,N
5  IPS(I)=I
   NM1=N-1
   DO 16 K=1,NM1
     RIF=0.0
     DO 11 I=K,N
       IP=IPS(I)
       IF(ABS(UL(IP,K)) .LE. RIF) GO TO 11
       RIF=ABS(UL(IP,K))
       IDXPIV=I
11  CONTINUE
       IF(IDXPIV .EQ. K) GO TO 15
       J=IPS(K)
       IPS(K)=IPS(IDXPIV)
       IPS(IDXPIV)=J
15  KP=IPS(K)
       PIVOT=UL(KP,K)
       KP1=K+1
       DO 16 I=KP1,N
         IP=IPS(I)
         EM=-UL(IP,K)/PIVOT
         UL(IP,K)=EM
         DO 16 J=KP1,N
           UL(IP,J)=UL(IP,J)+EM*UL(KP,J)
16  CONTINUE
       RETURN
     END

```

```

SUBROUTINE SOLVE(UL,X,B,N)
DIMENSION UL(80,80),B(80),X(80)
COMMON /SOL/IPS(80)
NP1=N+1
IP=IPS(1)
X(1)=B(IP)
DO 2 I=2,N
IP=IPS(I)
IM1=I-1
SUM=0.
DO 1 J=1,IM1
1 SUM=SUM+UL(IP,J)*X(J)
2 X(I)=B(IP)-SUM
IP=IPS(N)
X(N)=X(N)/UL(IP,N)
DO 4 IRACK=2,N
I=NP1-IRACK
IP=IPS(I)
IP1=I+1
SUM=0.0
DO 3 J=IP1,N
3 SUM=SUM+UL(IP,J)*X(J)
4 X(I)=(X(I)-SUM)/UL(IP,I)
RETURN
END

```

3.6 SUBROUTINE ELLI (M1, K, E, KPR, KEY)

This computes the elliptic integrals $K(m)$ and $E(m)$ and the derivative $K'(m)$ from their power series approximations (see Abramowitz and Stegun, 1964, p. 590).

Arguments :

M1	Real, the quantity $(1 - m)$.
K	Real, $K(m)$.
E	Real, $E(m)$.
KPR	Real, $K'(m)$.
KEY	Integer: 0 Compute K, E and KPR; 1 Compute K, E but omit KPR.

```

SUBROUTINE ELLI(M1,K,E,KPR,KEY)
REAL M1,K,KPR
T=-ALOG(M1)
K=1.386294+.5*T+M1*(9.666344E-2+.1249859*T+M1*(3.590092E-2
5 +6.880249E-2*T+M1*(3.742564E-2+3.328355E-2*T+M1*(1.451196E-2
5 +4.41787E-3*T))))
E=1.0+M1*(.4432514+.2499837*T+M1*(6.260601E-2+9.20018E-2*T+M1*(
8 4.757384E-2+4.069698E-2*T+M1*(1.736506E-2+5.264496E-3*T))))
IF(KEY .EQ. 1) RETURN
KPR=.5/M1 + 2.83225E-2 - .1249859*T + M1*(-2.999362E-3-.137605*T
6 +M1*(-7.899336E-2 - 9.985066E-2*T + M1*(-5.362998E-2 -
5 1.767148E-2 * T )))
RETURN
END

```

A.7 SUBROUTINE SETUP (KEYP11, KEYM11, I, J, L, API11,
API33, AMI11, IJ)

This is essential in computing the linear systems. Specifically SETUP, after calling ELLI, computes the quantities API11 (K_1 of eq. (8), API33 (K_0 of eq. 12), Ω_1 (eq. 20) and Ω_2 (eq. 21). The quantities Ω_1 and Ω_2 are used to compute $f(i, j)$ (AMI11) of eq. (39).

Arguments :

KEYP11	0 when computing API11 and API33, else 1.
KEYM11	0 when computing AMI11, else 1.
I	Subscript of observer (unprimed) cell.
J	Subscript of remote (primed) cell.
L	Index of the point within remote cell for which the kernels are to be computed (see DATA, Comments).
API11, API33 AMI11	Described above
IJ	0: use last value of M1 in kernel computations; 1: compute new M1.


```

SUBROUTINE SETUP(KEYP11,KEYM11,I,J,L,API11,API33,AMI11,IJ)
COMMON RHO(80,9),Z(80,9),ARC(80),C(80,9),S(80,9)
PEAL M,M1,K,KPR
ZD=Z(J,L)-Z(I,8)
R=RHO(I,8)
RP=RHO(J,L)
IF(IJ .EQ. 0) GO TO 115
RRP=R*RP
A1=RRP+RRP
A2=R*R+RP*RP+ZD*ZD
M1=(A2-A1)/(A2+A1)
M=1.-M1
CALL ELLI(M1,K,E,KPR,KEYM11)
A0=M/RRP
A1=SQRT(A0)
A2=M+M
A3=2.-M
IF(KEYP11-1)113,114,114
113 API11=A1*(A3*K-E-E)/M
API33=A1*K
114 A1=A1*A0
A3=.5*A3
IF(KEYM11-1)115,116,116
115 A0=C(J,L)
OM1=-A1*(.25*K-A3*KPR)
OM2=A1*(E-A3*((A3+M)*K - A2*A3*KPR))/(M*M)
AMI11=R*A0*OM2+(ZD*S(J,L)-RP*A0)*OM1
116 RETURN
END

```

REFERENCES

- Abramowitz, M. and I. A. Stegun (eds.) (1964), Handbook of Mathematical Functions, NBS Appl. Math. Series No. 55.
- Forsythe, G. and C.B. Moler (1967), Computer Solution of Linear Algebraic Systems, Prentice Hall, Englewood Cliffs.
- Keller, J.B., R.E. Kleinman and T.B.A. Senior (1972), Dipole Moments in Rayleigh Scattering, J. Inst. Math. and Applics., 9, 14-22.
- Senior, T. B. A. and D. J. Ahlgren (1972), The Numerical Solution of Low Frequency Scattering Problems, The University of Michigan Radiation Laboratory Report No. 013630-9-T.